

Frazioni egizie

- Note

- Autore

Claudio Marsan
Liceo Cantonale di Mendrisio
Via Agostino Maspoli
CH-6850 Mendrisio (Switzerland)
e-mail: claudio.marsan@liceomendrisio.ch

- Versione

Versione 2.0, 14 marzo 2003
Maple V Release 6.02 for Windows 2000

> **restart:**

Gli antichi egiziani non avevano simboli per rappresentare frazioni con il numeratore diverso da 1 (a parte per la frazione $2/3$, per la quale c'era un geroglifico particolare). Essi esprimevano dunque ogni frazione come somma di frazioni avente 1 a numeratore, per esempio: $5/6 = 1/2 + 1/3$.

La procedura seguente accetta come parametro un numero razionale positivo q e ritorna una lista di frazioni di numeratore 1, la cui somma dà q (se il numeratore di q è maggiore del denominatore di q , allora come primo elemento della lista verrà restituito un intero adeguato).

```
> frac_eg := proc(q :: rational )  
  local a, b, num, den, d0, d, n, h, lista;  
  a := numer(q);  
  b := denom(q);  
  num := a;  
  den := b;  
  if num = 1 then  
    d0 := b + 1  
  else  
    d0 := 2;  
  end if;  
  if a > b then  
    h := iquo(a, b);  
    num := a - h*b;  
    lista := [h];  
  else  
    lista := [];  
  end if;  
  if a = 0 then  
    n := 0  
  else
```

```

    n := 1;
end if;
while n <> 0 do
    d := d0;
    n := num/den - 1/d;
    if n >= 0 then
        lista := [op(lista), 1/d];
    else
        n := num/den;
    end if;
    num := numer(n);
    den := denom(n);
    d0 := d + 1;
end do;
lista;
end;

```

>

```
frac_eg := proc(q::rational)
```

```
local a, b, num, den, d0, d, n, h, lista;
```

```
    a := numer(q);
```

```
    b := denom(q);
```

```
    num := a;
```

```
    den := b;
```

```
    if num = 1 then d0 := b + 1 else d0 := 2 end if;
```

```
    if b < a then h := iquo(a, b); num := a - h*b; lista := [h] else lista := [ ] end if;
```

```
    if a = 0 then n := 0 else n := 1 end if;
```

```
    while n ≠ 0 do
```

```
        d := d0;
```

```
        n := num / den - 1 / d;
```

```
        if 0 ≤ n then lista := [op(lista), 1 / d] else n := num / den end if;
```

```
        num := numer(n);
```

```
        den := denom(n);
```

```
        d0 := d + 1
```

```
    end do;
```

```
    lista
```

```
end proc
```

11/12 = 1/2 + 1/3 + 1/12

```
> q := 11/12;
```

```
> F := frac_eg(q);
```

```
> somma := sum(F[j], j=1..nops(F));
```

$$q := \frac{11}{12}$$

$$F := \left[\frac{1}{2}, \frac{1}{3}, \frac{1}{12} \right]$$

$$\text{somma} := \frac{11}{12}$$

$$18/7 = 2 + 1/2 + 1/14$$

```
> q := 18/7;  
> F := frac_eg(q);  
> somma := sum(F[j],j=1..nops(F));
```

$$q := \frac{18}{7}$$

$$F := \left[2, \frac{1}{2}, \frac{1}{14} \right]$$

$$\text{somma} := \frac{18}{7}$$

$$1/137 = 1/138 + 1/18906$$

```
> q := 1/137;  
> F := frac_eg(q);  
> somma := sum(F[j],j=1..nops(F));  
> printf(`Nota:`); 138 = ifactor(138); 18906 = ifactor(18906);
```

$$q := \frac{1}{137}$$

$$F := \left[\frac{1}{138}, \frac{1}{18906} \right]$$

$$\text{somma} := \frac{1}{137}$$

Nota:

$$138 = (2) (3) (23)$$

$$18906 = (2) (3) (23) (137)$$

$$1 = 1/2 + 1/3 + 1/6$$

```
> q := 1;  
> F := frac_eg(q);  
> somma := sum(F[j],j=1..nops(F));
```

$$q := 1$$

$$F := \left[\frac{1}{2}, \frac{1}{3}, \frac{1}{6} \right]$$

$$\text{somma} := 1$$